15.2 Cumulative Distribution Function and Percentage Points for Normal Probability Distribution

A. Purpose

The procedures described in this chapter compute the Cumulative Distribution Function (CDF) and the percentage points of the Normal or Gaussian distribution. The CDF is sometimes called the lower tail. The lower tail, or CDF, $g(x;\mu,\sigma)$, and the upper tail, $h(x;\mu,\sigma)$ for the Normal probability distribution with mean μ and standard deviation σ are defined by

$$g(x;\mu,\sigma) = \frac{1}{\sigma\sqrt{2\pi}}\int_{-\infty}^{x} \exp\left(-\frac{(t-\mu)^2}{2\sigma^2}\right) dt,$$
$$h(x;\mu,\sigma) = \frac{1}{\sigma\sqrt{2\pi}}\int_{x}^{\infty} \exp\left(-\frac{(t-\mu)^2}{2\sigma^2}\right) dt = g(-x;\mu,\sigma)$$

The percentage point of a distribution is the value of x that gives the lower tail a specified value. In this case, the problem is to compute x given $u = g(x;\mu,\sigma)$, μ and σ, that is, compute $x = g^{-1}(u;\mu,\sigma)$.

B. Usage

B.1 Cumulative Distribution Function

B.1.a Program Prototype, Single Precision

REAL U, X, MU, SIGMA, SCDNML

EXTERNAL SCDNML

Assign values to X, MU and SIGMA and obtain U $= g(x;\mu,\sigma)$ by using

U = SCDNML(X, MU, SIGMA)

B.1.b Argument Definitions

X [in] Argument x of the function $g(x;\mu,\sigma)$.

MU [in] Parameter μ of the function $g(x;\mu,\sigma)$.

SIGMA [in] Parameter σ of the function $g(x;\mu,\sigma)$.

B.2 Percentage Points

B.2.a Program Prototype, Single Precision

REAL U, X, MU, SIGMA, SPPNML

EXTERNAL SPPNML

Assign values to U, MU and SIGMA and obtain X $= g^{-1}(u;\mu,\sigma)$ by using

X = SPPNML(U, MU, SIGMA)

B.2.b Argument Definitions

U [in] Argument u of the function $g^{-1}(u;\mu,\sigma)$. Require $0.0 \le \mathrm{U} \le 1.0$.

MU [in] Parameter μ of the function $g^{-1}(u;\mu,\sigma)$.

SIGMA [in] Parameter σ of the function $g^{-1}(u;\mu,\sigma)$.

B.3 Modifications for Double Precision

For double precision computation, change the REAL type statement to DOUBLE PRECISION and change the initial letter of the function names to D. Since these functions are not generic intrinsic functions, it is important to declare them explicitly to be DOUBLE PRECISION, because the default implicit type would be REAL.

C. Example and Remarks

See DRDCDNML and ODDCDNML for an example of the usage of these subprograms.

D. Functional Description

D.1 Method

To avoid cancellation error when $x-\mu << 0$, the identity $g(x;\mu,\sigma) = \frac{1}{2}\,\mathrm{erfc}((\mu-x)/\sigma\sqrt{2})$ is used. This expression never causes more cancellation error than mathematically equivalent alternatives, so it is used for all allowed values of x, μ, and σ (see Section E for restrictions). The procedure SERFC described in Chapter 2.2 is used to evaluate $\mathrm{erfc}((\mu-x)/\sigma\sqrt{2})$.

To compute the percentage points, invert the last expression to compute $x = g^{-1}(u;\mu,\sigma) = \mu - \sigma\sqrt{2}\,\mathrm{erfc}^{-1}(2u)$. The procedure SERFCI described in Chapter 2.13 is used to evaluate $\mathrm{erfc}^{-1}(2u)$.

D.2 Accuracy Tests

See Sections 2.2.D and 2.13.D.

E. Error Procedures and Restrictions

The procedure SERFC issues a warning message by way of the error message processor described in Chapter 19.2 if $(\mathrm{X}-\mathrm{MU})/(\sqrt{2.0}\times\mathrm{SIGMA}) < -xmax$. The value of $xmax$ depends on the system and the precision. Let $t = \sqrt{-\log(\sqrt{\pi}f)}$ where f is the underflow limit provided by R1MACH(1) or D1MACH(1) of Chapter 19.1. Then $xmax = t - ((\log t)/t) - 0.01$. For example, $xmax \approx 9.18$ (26.5) for single (double) precision IEEE arithmetic. The procedure SERFCI issues an error message at level 2 by way of the error message processor described in Chapter 19.2 if U < 0.0 or U > 1.0.

F. Supporting Information

Designed and programmed by W. V. Snyder, JPL, 1993.

Entry	Required Files
DCDNML	AMACH, DCSEVL, DERF, DERM1, DERV1, DINITS, ERFIN, ERMSG, IERM1, IERV1
DPPNML	AMACH, DERFI, DERM1, DERV1, DPPNML, ERFIN, ERMSG

Entry	Required Files
SCDNML	AMACH, ERFIN, ERMSG, IERM1, IERV1, SCSEVL, SERF, SERM1, SERV1, SINITS
SPPNML	AMACH, ERFIN, ERMSG, SERFI, SERM1, SERV1, SPPNML

DRDCDNML

```
      program DRDNML
c>> 2001-05-25 DRDCDNML Krogh Added comma to format.
c>> 1996-05-28 DRDCDNML Krogh  Changes to use M77CON
c>> 1994-07-06 DRDCDNML WV Snyder JPL set up for CHGTYP
c>> 1994-04-12 DRDCDNML WV Snyder JPL repair format to display sign
c
c     Evaluate the Cumulative Normal Distribution using DCDNML.
c
c--D replaces "?": DR?NML, DR?CDNML, ?CDNML, ?PPNML
      double precision DCDNML, DPPNML
      external DCDNML, DPPNML
      double precision X, C, P, MU, SIGMA
c
10    format ('         X          C=DCDNML(X)      DPPNML(C)')
20    format (1x,1p,2g14.7,2x,g14.7)
      x = -4.0d0
      mu = 0.0d0
      sigma = 1.0d0
      print 10
30    if (x .le. 4.0d0) then
         c = dcdnml(x,mu,sigma)
         p = dppnml(c,mu,sigma)
         print 20, x, c, p
         x = x + 0.5d0
         go to 30
      end if
      stop
      end
```

ODDCDNML

```
         X          C=DCDNML(X)      DPPNML(C)
 -4.000000      3.1671242E-05    -4.000000
 -3.500000      2.3262908E-04    -3.500000
 -3.000000      1.3498980E-03    -3.000000
 -2.500000      6.2096653E-03    -2.500000
 -2.000000      2.2750132E-02    -2.000000
 -1.500000      6.6807201E-02    -1.500000
 -1.000000      0.1586553        -1.000000
-0.5000000      0.3085375       -0.5000000
  0.000000      0.5000000         0.000000
 0.5000000      0.6914625        0.5000000
  1.000000      0.8413447         1.000000
  1.500000      0.9331928         1.500000
  2.000000      0.9772499         2.000000
  2.500000      0.9937903         2.500000
  3.000000      0.9986501         3.000000
  3.500000      0.9997674         3.500000
  4.000000      0.9999683         4.000000
```